
Data Types Comparison Table
    

This page contains a table that compares the data types used by Cassandra with their counterparts used by Xandra. Each Cassandra data type corresponds to one or more Elixir data types. What Elixir data type is used for a given Cassandra type is often controlled by a formatting option. For example, a date value from Cassandra can be returned as a Date.t() struct or as an integer representing the number of days since the Unix epoch depending on the :date_format option passed to various Xandra functions.

  
    
  
  Comparison Table


	Cassandra data type	Elixir data type
	ascii	String.t/0
	bigint	integer/0
	blob	binary/0
	boolean	boolean/0
	counter	integer/0
	date	Date.t/0 (if date_format: :date)
	date	integer/0 (if date_format: :integer), days from Unix epoch
	decimal	{value, scale} (if decimal_format: :tuple), where value * 10^(-1 * scale)
	decimal	Decimal.t/0 (if decimal_format: :decimal)
	double	float/0
	float	float/0
	inet	:inet.ip_address/0
	int	integer/0
	list<T>	list/0
	map<KeyT, ValueT>	map/0
	NULL	nil
	set<T>	MapSet.t/0
	smallint	integer/0
	text	String.t/0
	time	integer/0 (if time_format: :integer), as nanoseconds from midnight
	time	Time.t/0 (if time_format: :time)
	timestamp	DateTime.t/0 (if timestamp_format: :datetime)
	timestamp	integer/0 (if timestamp_format: :integer), as milliseconds from Unix epoch
	timeuuid	binary/0
	tinyint	integer/0
	tuple<...>	tuple/0
	uuid	binary/0
	varchar	String.t/0
	varint	integer/0



  

    
Compatibility
    

This page describes Xandra's compatibility with various Cassandra versions,
Cassandra-compatible data storage (like ScyllaDB), and Cassandra native protocol
versions.
All compatibility below refers to the latest version of Xandra.

  
    
  
  Cassandra Versions


	3.x
	4.x


  
    
  
  Native Protocol Versions


	v3
	v4
	v5


  
    
  
  ScyllaDB Versions


	2.3
	3.x
	4.x
	5.x



  

    
Telemetry Events
    


  
    
  
  Connection Events



  
    
  
  [:xandra, :connected]


Executed when a connection connects to its Cassandra node.
Measurements: none
Metadata:
	:connection (pid/0) - the PID of the connection process
	:connection_name (term/0 or nil) - given name of the connection or nil if not set
	:address (String.t/0) - the address of the node the connection is connected to
	:port (:inet.port_number/0) - the port of the node the connection is connected to
	:protocol_module (module/0) - the protocol module used for the connection
	:supported_options (map/0) - Cassandra supported options (mostly useful for internal debugging)


  
    
  
  [:xandra, :disconnected]


Executed when a connection disconnects from its Cassandra node.
Measurements: none
Metadata:
	:connection (pid/0) - the PID of the connection process
	:connection_name (term/0 or nil) - given name of the connection or nil if not set
	:address (String.t/0) - the address of the node the connection is connected to
	:port (:inet.port_number/0) - the port of the node the connection is connected to
	:reason (usually a Exception.t/0) - the reason for the disconnection


  
    
  
  [:xandra, :failed_to_connect]


Available since v0.18.0.
Executed when a connection fails to connect to its Cassandra node.
Measurements: none
Metadata:
	:connection (pid/0) - the PID of the connection process
	:connection_name (term/0 or nil) - given name of the connection or nil if not set
	:address (String.t/0) - the address of the node the connection is connected to
	:port (:inet.port_number/0) - the port of the node the connection is connected to
	:reason (usually a Exception.t/0) - the reason for the disconnection


  
    
  
  Query Events


The [:xandra, :prepare_query, ...] and [:xandra, :execute_query, ...] events are
Telemetry spans. See
telemetry:span/3. All the time
measurements are in native time unit, so you need to use System.convert_time_unit/3
to convert to the desired time unit.

  
    
  
  [:xandra, :prepare_query, ...]


Executed before and after a query is prepared (as a Telemetry span).
Measurements:
	:system_time (integer/0) - in :native time units (only for [..., :start] events)
	:monotonic_time (integer/0) - in :native time units
	:duration (integer/0) - in :native time units (only for [..., :stop] and [..., :exception] events)

Metadata:
	:connection (pid/0) - the PID of the connection process
	:connection_name (term/0 or nil) - given name of the connection or nil if not set
	:address (String.t/0) - the address of the node the connection is connected to
	:port (:inet.port_number/0) - the port of the node the connection is connected to
	:query (Xandra.Prepared.t/0) - the query being prepared
	:extra_metadata (any term) - extra metadata provided by the :telemetry_metadata option
	:reprepared (boolean/0) - whether the query was reprepared or not (only available for
[..., :stop] events)
	:reason (any term) - if there the result of the query was an error, this is the reason
(only available for [..., :stop] events), otherwise it's the error
that was raised (only available for [..., :exception] events)
	:kind (Exception.kind/0) - exception kind (only available for [..., :exception] events)
	:stacktrace (Exception.stacktrace/0) - exception stacktrace (only available for [..., :exception] events)


  
    
  
  [:xandra, :execute_query, ...]


Executed before and after a query is executed (as a Telemetry span).
Measurements:
	:system_time (integer/0) - in :native time units (only for [..., :start] events)
	:monotonic_time (integer/0) - in :native time units
	:duration (integer/0) - in :native time units (only for [..., :stop] and [..., :exception] events)

Metadata:
	:connection (pid/0) - the PID of the connection process
	:connection_name (term/0 or nil) - given name of the connection or nil if not set
	:address (String.t/0) - the address of the node the connection is connected to
	:port (:inet.port_number/0) - the port of the node the connection is connected to
	:query (Xandra.Simple.t/0, Xandra.Batch.t/0, or Xandra.Prepared.t/0) - the query being executed
	:extra_metadata (any term) - extra metadata provided by the :telemetry_metadata option
	:reason (any term) - if there the result of the query was an error, this is the reason
(only available for [..., :stop] events), otherwise it's the error
that was raised (only available for [..., :exception] events)
	:kind (Exception.kind/0) - exception kind (only available for [..., :exception] events)
	:stacktrace (Exception.stacktrace/0) - exception stacktrace (only available for [..., :exception] events)


  
    
  
  [:xandra, :prepared_cache, :hit | :miss]


Executed when a query is executed and the prepared cache is checked.
Measurements: none
Metadata:
	:connection (pid/0) - the PID of the connection process
	:connection_name (term/0 or nil) - given name of the connection or nil if not set
	:address (String.t/0) - the address of the node the connection is connected to
	:port (:inet.port_number/0) - the port of the node the connection is connected to
	:query (Xandra.Prepared.t/0) - the query being prepared
	:extra_metadata (any term) - extra metadata provided by the :telemetry_metadata option


  
    
  
  Warnings



  
    
  
  [:xandra, :server_warnings]


Executed when a query returns warnings.
Measurements:
	:warnings (non-empty list of String.t/0) - a list of warnings

Metadata:
	:address (String.t/0) - the address of the node the connection is connected to
	:port (:inet.port_number/0) - the port of the node the connection is connected to
	:current_keyspace (String.t/0 or nil) - the current keyspace of the connection, or nil if not set
	:query (Xandra.Simple.t/0, Xandra.Batch.t/0, or Xandra.Prepared.t/0) - the query that caused the warnings


  
    
  
  Cluster Events


Unless specified otherwise, these are available since v0.15.0.

  
    
  
  [:xandra, :cluster, :change_event]


Emitted when there is a change in the cluster, either as reported by Cassandra itself
or as detected by Xandra.
Measurements: none
Metadata:
	:cluster_pid (pid/0) - the PID of the cluster process
	:cluster_name (String.t/0 or nil) - the name of the cluster executing the event, if provided
through the :name option in Xandra.Cluster.start_link/1
	:host (Xandra.Cluster.Host.t/0) - the host the event is related to
	:event_type (atom/0) - one of :host_up (a host went up), :host_down (a host went down),
:host_added (a host was added to the cluster topology), or :host_removed
(a host was removed from the cluster topology)


  
    
  
  [:xandra, :cluster, :discovered_peers]


Available since v0.17.0.
Executed when the Xandra cluster's control connection discovers peers. The peers might have been
already discovered in the past, so you'll need to keep track of new peers if you need to.
Measurements:
	:peers (list of Xandra.Cluster.Host.t/0) - the discovered peers

Metadata:
	:cluster_pid (pid/0) - the PID of the cluster process
	:cluster_name (String.t/0 or nil) - the name of the cluster executing the event, if provided
through the :name option in Xandra.Cluster.start_link/1


  
    
  
  [:xandra, :cluster, :pool, :started | :restarted | :stopped]


Available since v0.17.0.
Executed when a connection pool to a node is started, restarted, or stopped.
Measurements: none
Metadata:
	:cluster_pid (pid/0) - the PID of the cluster process
	:cluster_name (String.t/0 or nil) - the name of the cluster executing the event, if provided
through the :name option in Xandra.Cluster.start_link/1
	:host (Xandra.Cluster.Host.t/0) - the host the event is related to


  
    
  
  [:xandra, :cluster, :control_connection, :connected]


Emitted when the control connection for the cluster is established.
Measurements: none
Metadata:
	:cluster_pid (pid/0) - the PID of the cluster process
	:cluster_name (String.t/0 or nil) - the name of the cluster executing the event, if provided
through the :name option in Xandra.Cluster.start_link/1
	:host (Xandra.Cluster.Host.t/0) - the host the event is related to


  
    
  
  [:xandra, :cluster, :control_connection, :disconnected | :failed_to_connect]


Emitted when the control connection for the cluster disconnects or fails to connect.
Measurements: none
Metadata:
	:cluster_pid (pid/0) - the PID of the cluster process
	:cluster_name (String.t/0 or nil) - the name of the cluster executing the event, if provided
through the :name option in Xandra.Cluster.start_link/1
	:host (Xandra.Cluster.Host.t/0) - the host the event is related to
	:reason (any term) - the reason for the disconnection or failure to connect



  

    
Xandra 
    



      
This module provides the main API to interface with Cassandra.
This module handles the connection to Cassandra, queries, connection pooling,
connection backoff, logging, and more.
db_connection
Before v0.18.0, this library was built on top of the
db_connection library. Since v0.18.0,
this is not the case anymore. This allowed us to significantly improve the design
and architecture of Xandra, ultimately resulting in a client that is more fitted
for the Cassandra protocol.


  
    
  
  Errors


Many of the functions in this module (whose names don't end with a !)
return values in the form {:ok, result} or {:error, error}. While result
varies based on the specific function, error is always one of the following:
	a Xandra.Error struct: such structs represent errors returned by
Cassandra. When such an error is returned, it means that communicating
with the Cassandra server was successful, but the server returned an
error. Examples of these errors are syntax errors in queries, non-existent
tables, and so on. See Xandra.Error for more information.

	a Xandra.ConnectionError struct: such structs represent errors in the
communication with the Cassandra server. For example, if the Cassandra
server dies while the connection is waiting for a response from the
server, a Xandra.ConnectionError error will be returned. See
Xandra.ConnectionError for more information.



  
    
  
  Parameters, encoding, and types


Xandra supports parameterized queries (queries that specify "parameter" values
through ? or :named_value):
SELECT * FROM users WHERE name = ? AND email = ?
SELECT * FROM users WHERE name = :name AND email = :email
When a query has positional parameters, parameters can be passed as a list to
functions like execute/4: in this case, a parameter in a given position in
the list will be used as the ? in the corresponding position in the
query. When a query has named parameters, parameters are passed as a map with
string keys representing each parameter's name and values representing the
corresponding parameter's value.

  
    
  
  Types


For information about how Elixir types translate to Cassandra types and
viceversa, see the "Data types comparison table" page.
Cassandra supports many types of values, and some types have "shades" that
cannot be represented by Elixir types. For example, in Cassandra an integer
could be a "bigint" (a 64 bit integer), an "int" (a 32 bit integer), a
"smallint" (a 16 bit integer), or others; in Elixir, however, integers are
just integers (with varying size to be precise), so it is impossible to
univocally map Elixir integers to a specific Cassandra integer type. For this
reason, when executing simple parameterized queries (statements) it is
necessary to explicitly specify the type of each value.
To specify the type of a value, that value needs to be provided as a
two-element tuple where the first element is the value's type and the second
element is the value itself. Types are expressed with the same syntax used in
CQL: for example, 16-bit integers are represented as "smallint", while maps
of strings to booleans are represented as "map<text, boolean>".
# Using a list of parameters:
statement = "INSERT INTO species (name, properties) VALUES (?, ?)"
Xandra.execute(conn, statement, [
  {"text", "human"},
  {"map<text, boolean>", %{"legs" => true, "arms" => true, "tail" => false}},
])

# Using a map of parameters:
statement = "INSERT INTO species (name, properties) VALUES (:name, :properties)"
Xandra.execute(conn, statement, %{
  "name" => {"text", "human"},
  "properties" => {"map<text, boolean>", %{"legs" => true, "arms" => true, "tail" => false}},
})
You only need to specify types for simple queries (statements): when using
prepared queries, the type information of each parameter of the query is
encoded in the prepared query itself.
# Using a map of parameters:
prepared = Xandra.prepare!(conn, "INSERT INTO species (name, properties) VALUES (:name, :properties)")
Xandra.execute(conn, prepared, %{
  "name" => "human",
  "properties" => %{"legs" => true, "arms" => true, "tail" => false},
})
User-defined types
Xandra supports user-defined types (UDTs). A UDT can be inserted as a map with
string fields. For example, consider having created the following UDTs:
CREATE TYPE full_name (first_name text, last_name text)
CREATE TYPE profile (username text, full_name frozen<full_name>)
and having the following table:
CREATE TABLE users (id int PRIMARY KEY, profile frozen<profile>)
Inserting rows will look something like this:
prepared_insert = Xandra.prepare!(conn, "INSERT INTO users (id, profile) VALUES (?, ?)")

profile = %{
  "username" => "bperry",
  "full_name" => %{"first_name" => "Britta", "last_name" => "Perry"},
}
Xandra.execute!(conn, prepared_insert, [_id = 1, profile])
Note that inserting UDTs is only supported on prepared queries.
When retrieved, UDTs are once again represented as maps with string
keys. Retrieving the row inserted above would look like this:
%{"profile" => profile} = conn |> Xandra.execute!("SELECT id, profile FROM users") |> Enum.fetch!(0)
profile
#=> %{"username" => "bperry", "full_name" => %{"first_name" => "Britta", "last_name" => "Perry"}}

  
    
  
  Values


Xandra supports two special values: nil and :not_set. Using nil explicitly
inserts a null value into the Cassandra table. This is useful to delete a value while
inserting. Note however that explicitly inserting null values into Cassandra creates so
called tombstones which negatively affects performance and resource utilisation and is
thus usually not recommended.
The :not_set value is a special value that allows to leave the value of a parametrized query
unset, telling Cassandra not to insert anything for the given field. In contrast to explicit
null values, no tombstone is created for this field. This is useful for prepared queries with
optional fields. The :not_set value requires Cassandra native protocol v4, available since
Cassandra 2.2.x. You can force the protocol version to v4 with the :protocol_version
option.

  
    
  
  Reconnections


A Xandra connection is able to handle connection
losses and to automatically reconnect to Cassandra. By default, reconnections
are retried at exponentially increasing randomized intervals, but backoff can
be configured through a subset of the options accepted by
start_link/2.

  
    
  
  Clustering


Xandra supports connecting to multiple nodes in a Cassandra cluster and
executing queries on different nodes based on load balancing strategies. See
the documentation for Xandra.Cluster for more information.

  
    
  
  Authentication


Xandra supports Cassandra authentication. See the documentation for
Xandra.Authenticator for more information.

  
    
  
  Retrying failed queries


Xandra takes a customizable and extensible approach to retrying failed queries
through "retry strategies" that encapsulate the logic for retrying
queries. See Xandra.RetryStrategy for documentation on retry strategies.

  
    
  
  Compression


Xandra supports compression. To inform the Cassandra server that the
connections you start should use compression for data transmitted to and from
the server, you can pass the :compressor option to start_link/1; this
option should be a module that implements the Xandra.Compressor
behaviour. After this, all compressed data that Cassandra sends to the
connection will be decompressed using this behaviour module.
To compress outgoing data (such as when issuing or preparing queries), the behavior
of Xandra depends on the native protocol that it negotiated with the Cassandra server.
For native protocol v4 and earlier, the :compressor option should be specified
explicitly for every request where you want outgoing compressor. When it's specified, the
given module will be used to compress data. If no :compressor option is
passed, the outgoing data will not be compressed. For native protocol v5 or later,
Xandra compresses all outgoing data since compression happens at the connection level,
so it's not necessary to pass the :compressor option when preparing or executing queries.
It's still good practice to pass it around so that your code stays agnostic of the
native protocol being used.

  
    
  
  Native protocol


Xandra supports the following versions of Cassandra's native protocol:
v5, v4, v3.
By default, Xandra will negotiate the protocol version with the Cassandra server.
Xandra will start by trying to use the highest protocol it supports (which is
v5). If the server rejects that, then
Xandra will reconnect with the protocol advertised by the server. If you
want to force a specific version of the native protocol that Xandra should use,
use the :protocol_version option.

  
    
  
  Logging


Xandra connections log a few events like disconnections or connection failures.
Logs contain the :xandra_address and :xandra_port metadata that you can
choose to log if desired.

  
    
  
  Custom payloads


The Cassandra native protocol supports exchanging custom payloads between
server and client. A custom payload is a map of string keys to binary values
(custom_payload/0).
To send custom payloads to the server, you can pass the :custom_payload
option to functions such as prepare/3 and execute/4.
If the server sends a custom payload in a response (result/0), you'll find
it in the :custom_payload field of the corresponding struct (such as
Xandra.Page, Xandra.Void, and so on).
By default, Cassandra itself ignores custom payloads sent to the server. Other
implementations built on top of the Cassandra native protocol might use
custom payloads to provide implementation-specific functionality. One such
example is Azure Cosmos DB.

    Xandra.Authenticator - Xandra v0.19.2
    
    

    


  
  

    
Xandra.Authenticator behaviour
    



      
A behaviour module for implementing a Cassandra authenticator.

  
    
  
  Examples


defmodule MyAuthenticator do
  @behaviour Xandra.Authenticator

  def response_body(options) do
    ["user:", Keyword.fetch!(options, :user), "_password:", Keyword.fetch!(options, :password)]
  end
end
To use the authenticator defined above:
Xandra.start_link(authentication: {MyAuthenticator, user: "foo", password: "bar"})
Xandra supports Cassandra's PasswordAuthenticator by default, see
Xandra.Authenticator.Password for more information.

      


      
        Summary


  
    Callbacks
  


    
      
        
  
    
    Xandra.Authenticator.Password - Xandra v0.19.2
    
    

    


  
  

    
Xandra.Authenticator.Password 
    



      
A Xandra.Authenticator that implements support for PasswordAuthenticator
to authenticate with Cassandra server.

  
    
  
  Example


options = [username: "xandra", password: "secret"]

Xandra.start_link(authentication: {Xandra.Authenticator.Password, options})

      




  

  
    
    Xandra.Batch - Xandra v0.19.2
    
    

    


  
  

    
Xandra.Batch 
    



      
Represents a batch of simple and/or prepared queries.
This module provides a data structure that can be used to group queries and
execute them as a Cassandra BATCH query. Batch queries can be executed
through Xandra.execute/3 and Xandra.execute!/3; see their respective
documentation for more information.
Note that the t/1 type is not documented as it's not meant
for public use. If you want to use batches in your typespecs, use t/0.

    Xandra.Cluster - Xandra v0.19.2
    
    

    


  
  

    
Xandra.Cluster 
    



      
Connection to a Cassandra cluster.
This module is a "proxy" connection with support for connecting to multiple
nodes in a Cassandra cluster and executing queries on such nodes based on a
given policy.

  
    
  
  Usage


This module manages pools of connections to different nodes in a Cassandra cluster.
Each pool is a pool of Xandra connections to a specific node.
The API provided by this module mirrors the API provided by the Xandra
module. Queries executed through this module will be "routed" to nodes
in the provided list of nodes based on a policy. See the
"Load balancing policies" section.
Regardless of the underlying pool, Xandra.Cluster will establish
one extra connection to a node in the cluster for internal purposes.
We refer to this connection as the control connection.
Here is an example of how one could use Xandra.Cluster to connect to a cluster:
Xandra.Cluster.start_link(
  nodes: ["cassandra1.example.net", "cassandra2.example.net"],
  pool_size: 10,
)
The code above will establish a pool of ten connections to each of the nodes
specified in :nodes, plus one extra connection used for internal
purposes, for a total of twenty-one connections going out of the machine.

  
    
  
  Child Specification


Xandra.Cluster implements a child_spec/1 function, so it can be used as a child
under a supervisor:
children = [
  # ...,
  {Xandra.Cluster, nodes: ["cassandra-seed.example.net"]}
]

  
    
  
  Contact Points and Cluster Discovery


Xandra.Cluster auto-discovers peer nodes in the cluster, by using the system.peers
built-in Cassandra table. Once Xandra discovers peers, it opens a pool of connections
to a subset of the peers based on the chosen load-balancing policy (see below).
The :nodes option in start_link/1 specifies the contact points. The contact
points are used to discover the rest of the nodes in the cluster. It's generally
a good idea to provide multiple contacts points, so that if some of those are
unreachable, the others can be used to discover the rest of the cluster. Xandra.Cluster
tries to connect to contact points in the order they are specified in the :nodes
option, initially ignoring the chosen load-balancing policy. Once a connection is
established, then that contact point is used to discover the rest of the cluster
and open connection pools according to the load-balancing policy.
Xandra also refreshes the cluster topology periodically. See the
:refresh_topology_interval option in start_link/1.

  
    
  
  Load-balancing Policies


Xandra.Cluster uses customizable "load-balancing policies" to manage nodes
in the cluster. A load-balancing policy is a module that implements the
Xandra.Cluster.LoadBalancing behaviour. Xandra uses load-balancing policies
for these purposes:
	Choosing which node to execute a query on
	Choosing which nodes to open pools of connections to (see the :target_pools
option in start_link/1)
	Choosing which node the control connection connects to (or re-connects to
in case of disconnections)

Xandra ships with the following built-in load-balancing policies:
	Xandra.Cluster.LoadBalancingPolicy.Random - it will choose one of the
connected nodes at random and execute the query on that node.
	Xandra.Cluster.LoadBalancingPolicy.DCAwareRoundRobin - it will execute the
queries on the nodes in a round robin manner, prioritizing the current DC.


  
    
  
  Disconnections and Reconnections


Xandra.Cluster also supports nodes disconnecting and reconnecting: if Xandra
detects one of the nodes in :nodes going down, it will not execute queries
against it anymore, but will start executing queries on it as soon as it
detects such node is back up.
If all specified nodes happen to be down when a query is executed, a
Xandra.ConnectionError with reason {:cluster, :not_connected} will be
returned.

  
    
  
  Telemetry


Xandra.Cluster emits several Telemetry events to help you log, instrument,
and debug your application. See the Telemetry Events
page in the guides for a comprehensive list of the events that Xandra emits.

    Xandra.Cluster.Host - Xandra v0.19.2
    
    

    


  
  

    
Xandra.Cluster.Host 
    



      
The data structure to represent a host in a Cassandra cluster.
The fields of this struct are public. See t/0 for information on their type,
and %Xandra.Cluster.Host{} for more information.

    Xandra.Cluster.LoadBalancingPolicy - Xandra v0.19.2
    
    

    


  
  

    
Xandra.Cluster.LoadBalancingPolicy behaviour
    



      
A behaviour for defining load-balancing policies.
A load-balancing policy is a way to tell Xandra which nodes to query when connected
to a cluster.
For example, a simple load-balancing policy is Xandra.Cluster.LoadBalancingPolicy.Random,
which picks a random node for each query out of the ones that are up.

    Xandra.Cluster.LoadBalancingPolicy.DCAwareRoundRobin - Xandra v0.19.2
    
    

    


  
  

    
Xandra.Cluster.LoadBalancingPolicy.DCAwareRoundRobin 
    



      
A Xandra.Cluster.LoadBalancingPolicy that prefers hosts in a "local" data center.
This policy uses a round-robin strategy to pick hosts, giving precedence to hosts
in a "local" data center. The local data center is determined by the
:local_data_center option (see below). "Giving precedence" means that
hosts_plan/1 and query_plan/1 will return a list of hosts where first there are all
the local hosts that are up, and then all the remote hosts that are up.
The round-robin strategy is applied to local and remote hosts separately. For example,
say the local hosts are LH1, LH2, LH3, and the remote hosts are RH1, RH2, RH3.
The first time, this policy will return [LH1, RH1, LH2, RH2, LH3, RH3]. The second time,
it will return [LH2, LH3, LH1, RH2, RH3, RH1]. And so on.
This policy is available since Xandra v0.15.0.

  
    
  
  Options


This policy supports the following initialization options:
	:local_data_center - The local data center. If :from_first_peer, this policy picks the local
data center from the first peer that is added to the cluster. This is
usually the first peer that the control connection connects to, which
(if everything goes well) is the first peer listed in the :nodes
option passed to Xandra.Cluster.start_link/1. If you want to force
a specific data center to be used as the local data center, you can
pass a string as the value of this option (such as "datacenter1"). The default value is :from_first_peer.

Xandra.Cluster.LoadBalancingPolicy.Random 
    



      
A simple Xandra.Cluster.LoadBalancingPolicy that picks hosts at random.
This load-balancing policy doesn't make any attempt to be smart: it doesn't take
data center or tokens into consideration, and considers all available nodes.
This policy is available since Xandra v0.15.0.

Xandra.Compressor behaviour
    



      
A behaviour to compress and decompress binary data.
Modules implementing this behaviour can be used to compress and decompress
data using one of the compression algorithms supported by Cassandra (see below).

  
    
  
  Supported algorithms and implementations


Native protocol versions v4 and earlier support two compression algorithms:
LZ4 and
Snappy.
Native protocol versions v5 and later only support LZ4.

  
    
  
  LZ4


If you implement a compressor module for the LZ4 algorithm, then:
	The compress/1 callback must return the compressed payload
preceded by a 32-bit big-endian unsigned integer representing the
length (in bytes) of the uncompressed body.

	Xandra will call the decompress/1 callback with the compressed
payload, also preceded by the uncompressed body size (in bytes) as
a 32-bit big-endian unsigned integer.


That is, the result of compression when using the LZ4 algorithm must look like
this:
<<uncompressed_payload_length::32-big-unsigned, compressed_payload::binary>>
Snappy is self-sufficient so it doesn't need the prepended uncompressed-payload
size.

  
    
  
  Example


Let's imagine you implemented the LZ compression algorithm in your application:
defmodule MyApp.LZ4 do
  def compress(binary), do: # ...
  def decompress(binary, uncompressed_size), do: # ...
end
You can then implement a module that implements the Xandra.Compressor
behaviour and can be used to compress and decompress data flowing through the
connection to Cassandra:
defmodule LZ4XandraCompressor do
  @behaviour Xandra.Compressor

  @impl true
  def algorithm(), do: :lz4

  @impl true
  def compress(body) do
    [<<IO.iodata_length(body)::4-unit(8)-integer>>, MyApp.LZ4.compress(body)]
  end

  @impl true
  def decompress(<<uncompressed_size::4-unit(8)-integer, compressed_body::binary>>) do
    MyApp.LZ4.decompress(compressed_body, uncompressed_size)
  end
end
Now, this module can be passed as the value of the :compressor option to
many functions in Xandra:
Xandra.start_link(compressor: LZ4XandraCompressor)
For more information on compression, see the "Compression" section in the
documentation for Xandra.

Xandra.Page 
    



      
A struct that represents a page of rows.
This struct represents a page of rows that have been returned by the
Cassandra server in response to a query such as SELECT, but have not yet
been parsed into Elixir values.
This struct implements the Enumerable protocol and is therefore a stream. It
is through this protocol that a Xandra.Page struct can be parsed into Elixir
values. The simplest way of getting a list of single rows out of a
Xandra.Page struct is to use something like Enum.to_list/1. Each element
emitted when streaming out of a Xandra.Page struct is a map of string column
names to their corresponding value.
See %Xandra.Page{} for information about which fields are public.

  
    
  
  Examples


statement = "SELECT name, age FROM users"

%Xandra.Page{} = page = Xandra.execute!(conn, statement, _params = [])

Enum.each(page, fn %{"name" => name, "age" => age} ->
  IO.puts("Read user with name #{name} (age #{age}) out of the database")
end)

Xandra.Prepared 
    



      
A data structure used to internally represent prepared queries.
See %Xandra.Prepared{} for information about which fields are
public. All other fields are documented in t/0 to avoid Dialyzer warnings,
but are not meant to be used by users.

Xandra.RetryStrategy behaviour
    



      
A behaviour that handles how to retry failed queries.
This behaviour makes it possible to customize the strategy that Xandra uses to
retry failed queries. By default, Xandra does not retry failed queries, and
does not provide any default retry strategy since retrying queries based on
the failure reason is very tied to application logic.
Why Do You Need This?
You might be wondering why there's a need for a retry strategy behaviour, when
you could do this at the application layer by simply checking the return value
of Xandra calls and potentially retrying.
Well, the reason is that retrying queries to Cassandra can get quite smart. For
example, you might want context to know what Cassandra node a query failed to
execute on, so that you can try it on a different node next. Xandra.RetryStrategy
modules get all the necessary info to implement this kind of smarter strategies.

A module that implements the Xandra.RetryStrategy behaviour can be passed to
several functions in the Xandra module and Xandra.Cluster modules. look at
the documentation for those modules for more information.

  
    
  
  Usage


When a query fails and a retry strategy module was passed as an option, Xandra
will:
	Invoke the new/1 callback — it will invoke this with the options
passed to the failing function to initialize the given retry strategy.
This gives you access to things like default consistency, timeouts, and
so on.

	Invoke the retry/3 callback until necessary — Xandra will ask the
retry strategy whether to retry or error out until either
the query succeeds or retry/3 says to error out.


Process
The new/1 and retry/3 callbacks will be invoked in the same
process that executed the original query.


  
    
  
  Single Connections or Clusters


There are two possible cases where a retry strategy is invoked: either it's invoked
when a query fails to execute on a single connection (that is, it was executed through
a Xandra function), or when a query fails to execute through a cluster connection (that
is, it was executed through a Xandra.Cluster function).
To distinguish these cases, Xandra always passes the :execution_level option
to new/1. This option has the type execution_level/0.
If the execution level is :single_connection, Xandra doesn't inject any additional
options. When the execution level is :single_connection, retry/3 can only return
the 3-element version of the {:retry, ...} tuple.
If the execution level is :cluster, Xandra injects these options when calling new/1:
	:connected_hosts — a list of {connection_pid, host} tuples, where
connection_pid (a pid/0) is the PID of the connection and host (a
Xandra.Cluster.Host.t/0) is the corresponding host information. You can use
this option to determine on which node to retry a query. Elements in this list
are ordered according to the Xandra.Cluster.LoadBalancingPolicy used by the
cluster. If you want to keep track of the original :connected_hosts, you'll
need to store them in the state of the retry strategy returned by new/1.

When the execution level is :single_connection, retry/3 can only return
the 4-element version of the {:retry, ...} tuple.

  
    
  
  Examples


Let's look at some examples.

  
    
  
  Retry Count


This is an example of a retry strategy that retries a fixed number of times
before failing. It injects :retry_count option which it uses to keep track
of how many times the query failed. This is effectively the state/0 of this
retry strategy.
defmodule MyApp.CounterRetryStrategy do
  @behaviour Xandra.RetryStrategy

  @impl true
  def new(options) do
    # This is the "state" of this retry strategy
    Keyword.fetch!(options, :retry_count)
  end

  @impl true
  def retry(error, options, retries_left)

  def retry(_error, _options, _retries_left = 0) do
    :error
  end

  def retry(_error, options, retries_left = _state) do
    {:retry, options, retries_left - 1}
  end
end

  
    
  
  Downgrading Consistency


Another interesting example could be a retry strategy based on downgrading
consistency: for example, we could execute all queries with a "high" consistency
(such as :all) at first, and in case of failure, try again with a lower
consistency (such as :quorum), finally giving up if that fails as well.
defmodule MyApp.DowngradingConsistencyRetryStrategy do
  @behaviour Xandra.RetryStrategy

  @impl true
  def new(_options) do
    :no_state
  end

  @impl true
  def retry(_error, options, :no_state) do
    case Keyword.fetch(options, :consistency) do
      # No consistency was specified, so we don't bother to retry.
      :error -> :error

      # If the consistency was :all, we downgrade it by injecting a new one in the options.
      {:ok, :all} -> {:retry, Keyword.replace!(options, :consistency, :quorum), :no_state}

      # If the consistency was already lower than :all, we give up and stop retrying.
      {:ok, _other} -> :error
    end
  end
end

  
    
  
  Different Node (for Clusters)


A particularly-useful application of retry strategies is to retry queries on different hosts
when using Xandra.Cluster. We can even choose not to execute on certain hosts
(because they may be in a different data center). The following example retries on all hosts
after the first :connected_node has failed:
defmodule MyApp.AllNodesRetryStrategy do
  @behaviour Xandra.RetryStrategy

  alias Xandra.Cluster.Host

  @impl true
  def new(options) do
    if options[:execution_level] != :cluster do
      raise ArgumentError, "this retry strategy can only be used with clusters"
    end

    [_already_tried_node | remaining_nodes] = Keyword.fetch!(options, [:connected_hosts])
    remaining_nodes
  end

  @impl true
  def retry(error, options, nodes)

  # No nodes left to retry on.
  def retry(_error, options, [] = _remaining_nodes) do
    :error
  end

  def retry(_error, options, [{conn_pid, %Host{}} | remaining_nodes]) do
    {:retry, options, _new_state = remaining_nodes, conn_pid}
  end
end

Xandra.SchemaChange 
    



      
A struct that represents the result of a query that modifies the schema.
See %Xandra.SchemaChange{} for information on the fields of this struct.

Xandra.SetKeyspace 
    



      
A struct that represents the result of a USE query.
These are the public fields of this struct:
	:keyspace - the keyspace (as a binary) that was set through the executed
USE query.

	:tracing_id - the tracing ID (as a UUID binary) if tracing was enabled,
or nil if no tracing was enabled. See the "Tracing" section in Xandra.execute/4.

	:custom_payload - the custom payload sent by the server, if present.
If the server doesn't send a custom payload, this field is nil. Otherwise,
it's of type Xandra.custom_payload/0. See the "Custom payloads" section
in the documentation for the Xandra module.

Xandra.Simple 
    



      
A data structure used internally to represent simple queries.

Xandra.Telemetry 
    



      
Telemetry integration for event tracing, metrics, and logging.
Xandra uses telemetry for reporting
metrics and events. Below we list all the possible events emitted by Xandra, alongside
their measurements and metadata.
Xandra emits telemetry events since v0.15.0.

  
    
  
  Events


For a comprehensive list of the events that Xandra emits, see the
Telemetry Events page in the guides.

Xandra.Void 
    



      
A struct that represents an empty Cassandra result.
This struct is returned as the result of queries such as INSERT, UPDATE, or
DELETE. See %Xandra.Void{} for information about the fields.

Xandra.ConnectionError exception
    



      
An exception struct that represents an error in the connection to the
Cassandra server.
For more information on when this error is returned or raised, see the
documentation for the Xandra module.
The :action field represents the action that was being performed when the
connection error occurred. The :reason field represents the reason of the
connection error: for network errors, this is usually a POSIX reason (like
:econnrefused). The following Xandra-specific reasons are supported:
	{:unsupported_compression, algorithm} - this happens when a
:compressor module has been specified in Xandra.start_link/1, but
negotiating the connection algorithm fails because such compressor module
uses an algorithm that the Cassandra server does not support.

	:disconnected - the connection closed in the middle of a request to the server.

	{:connection_process_crashed, reason} - the connection process crashed before
sending a response.

	:timeout - the connection timed out while waiting for a response from the
server.

	{:cluster, :not_connected} - this happens when a Xandra.Cluster-based
connection is not connected to any node (for example, because all the
specified nodes are currently down). See the documentation for
Xandra.Cluster for more information.


Since this struct is an exception, it is possible to raise it with
Kernel.raise/1. If the intent is to format connection errors as strings (for
example, for logging purposes), it is possible to use Exception.message/1 to
get a formatted version of the error.

